

中華民國第42屆中小學科學展覽會

::: 作品說明書 :::

高職組-電子

科 別：資 訊

組 別：高 職 組

作品名稱：電腦守護天使

關 鍵 詞：8051、89C51

編 號：091005

學校名稱：

臺北縣私立復興高級商工職業學校

作者姓名：

戴百賢、陳金生

指導老師：

謝建立




壹、摘要

這次的製作電腦守護天使，會讓我們了解了解到要維護電腦的安全並不是一件容易的事，但卻是一件相當重要且刻不容緩的事。

你想以為說只要有掃毒程式就可以預防的非常徹底嗎？

也學到了「說的到不一定就做得得到」 我們原本預計的期望是更高的，雖然學藝尚未完全成熟，但我們相信只要全組的組員都能夠盡全力的製作、相輔相成的心，以自己的長處來彌補別人不足的部分，使我們發輝團結的精神，相信一定能夠創造出最佳的成績。

而我們未來的期望是要將「電腦守護天使」能加以商品化後在加強其功能性，使以便更多的電腦使用者，可以放心的安裝自己專屬的個人電腦。



貳、研究動機

雖然現今的電腦廠商不斷的推出一些資料的保護措施，用來防止宵小之人使用、竊取甚至是破壞電腦中的寶貴資料，但是只有這些防護措施是不夠的，因為不管你是如何的設定密碼，任何懂得較高等電腦知識的人只需要動一點小腦筋，還是有可能將它給破解的，所以現在我們這組所聯想到，就是多一層的保護措施那就是利用在學校課堂上所學到的單晶片程式設計在開機之前多一個防護，以達到防止寶貴資料的流失或被竊取。這次的製作中我們學到了很多也用到了一些以前沒有用過的電子零件（8051 單晶片實作入門與應用）和電腦裡面一些從未見過的主機板零件（電腦硬體裝修），以及搭配單晶片的程式編碼（數位邏輯設計）燒錄、模擬等。

參、研究目的


之所以叫「手」護神，事實上是運用單晶片來設計一個電子的密碼鎖，以達到控制各項電器設備為主，而這電子式的「手」護神，算是多功能的設計，平時是電子鐘用來顯示時間、又可以當成各類保護系統的運用、對於不想讓小孩一直打電動的家長也有幫助，也許會對將來會有所貢獻，也許會被改良更好的鎖，讓當之的小人，萬嘆有此發明可以應用在保險櫃、金庫... ..等等。簡單的說使用電腦的人，必須先輸入一組自己設定的密碼，才能使電源導通然後傳入電腦主機才可開機，以達到多一層的保護措施。

肆、研究設備及器材

名稱	型號	單位	數量	備註
IC	89C51	個	1	附 IC 座
IC	93C66	個	1	附 IC 座
LCD	16 x 2	個	1	
LED	小圓點	顆	1	
電阻	220 Ω 1/4w	個	1	
電阻	10k Ω 1/4w	個	5	
電解電容	10 μ F/16v	個	1	
陶瓷電容	30P	個	2	
石英晶體	12MHz	個	1	
排線	14、4pin	條	2	
電路板	300 x 600	塊	2	
電腦主機設備	Pentium 級	套	1	
無段按鍵	圓形	個	16	

伍、研究過程或方式


一、 流程圖




伍、研究過程或方式

二、外觀圖

電路實體部分




控制器和 LCD 顯示器部分



伍、研究過程或方式

三、電路圖



伍、研究過程或方式

四、操作說明

「HELP」鍵：按鍵功能顯示

「F1」鍵：顯示時間

「F2」鍵：調整時間

調整步驟：

- 1、按下「F2」
- 2、輸入 6 個數字，代表時、分、秒
- 3、按下「SET」

「F3」鍵：輸入密碼

輸入步驟：

- 1、按下「F3」
- 2、輸入 4 個密碼數字
- 3、按下「SET」

若密碼正確會顯示「EXACT」，LED 亮

若密碼錯誤會顯示「ERROR」

「F4」鍵：重設密碼

設定步驟：

- 1、按下「F4」
- 2、輸入 4 個密碼數字
- 3、按下「SET」

若密碼設定完成會顯示「SUCCESS！」

為因應商品化會將此功能取消

伍、研究過程或方式

五、程式編碼

```

EN      REG    P2.7    ;LCD 模組致能腳
RW      REG    P2.6    ;LCD 模組讀寫控制腳
RS      REG    P2.5    ;LCD 模組資料/指令選擇腳
SW      REG    P2.4    ;LED 控制腳(密碼鎖控制)
CS      REG    P2.3    ;93C66 晶片選擇腳
SK      REG    P2.2    ;93C66 串列時脈輸入腳
DI      REG    P2.1    ;93C66 串列資料輸入腳
DO      REG    P2.0    ;93C66 串列資料輸出腳
ADDR    EQU    37H     ;93C66 讀寫位址
CNT      EQU    38H
SEC      EQU    39H     ;計時器 0 中斷次數
;
      ORG      00H
      AJMP     START
      ORG      0BH
      AJMP     TIMER0
;-----
; MAIN
;-----
START:  MOV     SP,#60H    ;改變堆疊指標
      CLR      SK         ;CK=0
      SETB     SW         ;關閉 LED(上鎖)
      ACALL    INIT       ;LCD 初始化
      ACALL    SET0       ;設定計數器 0
L0:     ACALL    ON        ;開機顯示功能表
L1:     ACALL    CONV      ;時間轉換，BCD 轉成 ASCII
      ACALL    SCAN       ;掃描鍵盤
      MOV      A,R5        ;鍵值 R5 存入累加器
      CJNE     A,#BH,L2;按下(HELP)鍵?
      AJMP     HELP
L2:     CJNE     A,#CH,L3;按下(F1)鍵?
      AJMP     F1
L3:     CJNE     A,#DH,L4    ;按下(F2)鍵?
      AJMP     F2
L4:     CJNE     A,#EH,L5;按下(F3)鍵?
      AJMP     F3
L5:     CJNE     A,#FH,L1;按下(F4)鍵?
      AJMP     F4
;-----
; HELP

```

```

;-----
HELP:  SETB   SW           ;關閉 LED(上鎖)
       SJMP   L0

;----
; F1;顯示時間
;----
F1:    ACALL  CLRLCD       ;清除 LCD 螢幕
       MOV    A,#86H
       ACALL  W_INS
       MOV    CNT,#4
       MOV    DPTR,#MSG3 ;顯示'TIME'
       ACALL  W_STR
F11:   ACALL  CONV         ;時間轉換，BCD 轉成 ASCII
       MOV    A,#C4H
       ACALL  W_INS
       ACALL  DISP         ;顯示時間
       ACALL  SCAN         ;掃描鍵盤
       MOV    A,R5          ;鍵值 R5 存入累加器
       CJNE   A,#BH,F12    ;按下(HELP)鍵?
       AJMP   HELP
F12:   CJNE   A,#DH,F13    ;按下(F2)鍵?
       AJMP   F2
F13:   CJNE   A,#EH,F14    ;按下(F3)鍵?
       AJMP   F3
F14:   CJNE   A,#FH,F11    ;按下(F4)鍵?
       AJMP   F4

;----
; F2;調整時間
;----
F2:    ACALL  CLRLCD       ;清除 LCD 螢幕
       MOV    R0,#48H      ;所設定時間之儲存緩衝區
       MOV    A,#82H
       ACALL  W_INS
       MOV    CNT,#11
       MOV    DPTR,#MSG4 ;顯示'ADJUST TIME'
       ACALL  W_STR
       MOV    A,#C4H
       ACALL  W_INS
F21:   ACALL  SCAN         ;掃描鍵盤
       MOV    A,R5
       XRL    A,#16        ;判斷是否有按下任何鍵?
       JZ     F21          ;若否，則重新掃描鍵盤
       MOV    A,R5          ;若是，將鍵值存入 R5 中
       CJNE   A,#AH,F22    ;按下(SET)鍵?
       AJMP   SETF2
F22:   CJNE   A,#BH,F23    ;按下(HELP)鍵?

```

```

        AJMP    HELP
F23: CJNE    A,#CH,F24    ;按下(F1)鍵?
        AJMP    F1
F24: CJNE    A,#DH,F25    ;按下(F2)鍵?
        AJMP    F21
F25: CJNE    A,#EH,F26    ;按下(F3)鍵?
        AJMP    F3
F26: CJNE    A,#FH,F27    ;按下(F4)鍵?
        AJMP    F4
F27: MOV     A,R5          ;按下數字鍵 0~9 , 顯示並儲存於緩衝區
        MOV     @R0,A
        ORL     A,#30H
        ACALL   W_DATA
        INC     R0
        CJNE    R0,#4EH,F28 ;緩衝區已滿?
        MOV     R0,#48H    ;緩衝區已滿 , 回至緩衝區開頭
        MOV     A,#C4H
        ACALL   W_INS
F28: SJMP    F21
;-----
; F2-SET;設定時間
;-----
SETF2: MOV     R2,#3
        MOV     R0,#48H
        MOV     R1,#33H
T1:     MOV     A,@R0
        ANL     A,#0FH      ;將位址 48H~4DH 之 ASCII 時間 ,
        SWAP    A           ;轉成 BCD 時間 , 並存入位址 33H~35H 中
        INC     R0
        ADD     A,@R0
        MOV     @R1,A
        INC     R0
        INC     R1
        DJNZ    R2,T1
;
        MOV     R0,#30H      ;'顯示時間'之位址為 30H~32H
        MOV     R1,#33H      ;'設定時間'之位址為 33H~35H
        MOV     A,@R1
        CLR     C
        SUBB    A,#24H        ;所設定時間之時>24?
        JNC     T2           ;若時>24 , 表示設定不合法
        MOV     A,@R1
        MOV     @R0,A
        INC     R0
        INC     R1
        MOV     A,@R1

```

```

CLR      C
SUBB     A,#60H      ;所設定時間之分>60?
JNC      T2          ;若分>60，表示設定不合法
MOV      A,@R1
MOV      @R0,A
INC      R0
INC      R1
MOV      A,@R1
CLR      C
SUBB     A,#60H      ;所設定時間之秒>60?
JNC      T2          ;若秒>60，表示設定不合法
MOV      A,@R1
MOV      @R0,A
T2:      AJMP      F1
;-----
; F3:輸入密碼
;-----
F3:      ACALL     CLRLCD      ;清除 LCD 螢幕
MOV      A,#83H
ACALL     W_INS
MOV      CNT,#10
MOV      DPTR,#MSG5 ;顯示'INPUT CODE'
ACALL     W_STR
MOV      A,#C6H
ACALL     W_INS
MOV      R0,#50H
F31:      ACALL     SCAN
MOV      A,R5
XRL      A,#16      ;是否按下任何按鍵?
JZ        F31
MOV      A,R5      ;按鍵已按下，鍵值存入 R5
CJNE     A,#AH,F32  ;按下(SET)鍵?
AJMP      SETF3
F32:      CJNE     A,#BH,F33  ;按下(HELP)鍵?
AJMP      HELP
F33:      CJNE     A,#CH,F34  ;按下(F1)鍵?
AJMP      F1
F34:      CJNE     A,#DH,F35  ;按下(F2)鍵?
AJMP      F2
F35:      CJNE     A,#EH,F36  ;按下(F3)鍵?
AJMP      F31
F36:      CJNE     A,#FH,F37  ;按下(F4)鍵?
AJMP      F4
F37:      MOV      A,R5      ;按鍵為數字鍵 0~9，鍵值存入 R5 中
MOV      @R0,A
ORL      A,#30H

```

```

        ACALL W_DATA      ;顯示鍵值並儲存至緩衝區
        INC      R0
        CJNE     R0,#54H,F38 ;緩衝區已滿?
        MOV      R0,#50H      ;緩衝區已滿，重新設定至緩衝區開頭
        MOV      A,#C6H
        ACALL    W_INS
F38: SJMP      F31
;-----
; F3-SET;輸入密碼
;-----
SETF3:  CLR      F0          ;F0=0，密碼錯誤。F0=1，密碼正確
        ACALL    R_CODE      ;自 93C66 讀取 4 位元組密碼
        MOV      R2,#4
        MOV      R0,#50H
        MOV      R1,#54H
C1:     MOV      A,@R0        ;比對密碼是否正確?
        XRL      A,@R1
        JNZ      C2
        INC      R0
        INC      R1
        DJNZ     R2,C1
        CLR      SW          ;密碼正確，開鎖
        SETB     F0
C2:     MOV      A,#C6H
        ACALL    W_INS
        MOV      DPTR,#MSG6 ;若密碼正確，顯示'EXACT'
        JB       F0,C3
        MOV      DPTR,#MSG7 ;若密碼錯誤，顯示'ERROR'
C3:     MOV      CNT,#5
        ACALL    W_STR
C4:     ACALL    SCAN         ;掃描按鍵
        MOV      A,R5
        CJNE     A,#BH,C5     ;按下(HELP)鍵?
        AJMP     HELP
C5:     CJNE     A,#CH,C6     ;按下(F1)鍵?
        AJMP     F1
C6:     CJNE     A,#DH,C7     ;按下(F2)鍵?
        AJMP     F2
C7:     CJNE     A,#EH,C8;按下(F3)鍵?
        AJMP     F3
C8:     CJNE     A,#FH,C9;按下(F4)鍵?
        AJMP     F4
C9:     AJMP     C4
;----
; F4;設定密碼
;----

```

```

F4:      ACALL  CLRLCD      ;清除 LCD 螢幕
        MOV     A,#84H
        ACALL  W_INS
        MOV     CNT,#8
        MOV     DPTR,#MSG8 ;顯示'SET CODE'
        ACALL  W_STR
        MOV     A,#C6H
        MOV     R0,#54H
        ACALL  W_INS
F41: ACALL  SCAN      ;掃描鍵盤
        MOV     A,R5
        XRL     A,#16      ;是否按下任何鍵?
        JZ      F41
        MOV     A,R5
        CJNE    A,#AH,F42   ;按下(SET)鍵?
        AJMP    SETF4
F42: CJNE    A,#BH,F43   ;按下(HELP)鍵?
        AJMP    HELP
F43: CJNE    A,#CH,F44   ;按下(F1)鍵?
        AJMP    F1
F44: CJNE    A,#DH,F45   ;按下(F2)鍵?
        AJMP    F2
F45: CJNE    A,#EH,F46   ;按下(F3)鍵?
        AJMP    F3
F46: CJNE    A,#FH,F47   ;按下(F4)鍵?
        AJMP    F4
F47: MOV     A,R5      ;鍵值為數字鍵，存入 R5 中
        MOV     @R0,A    ;將鍵值存入緩衝區
        ADD     A,#30H
        ACALL  W_DATA
        INC     R0
        CJNE    R0,#58H,F48 ;緩衝區已滿?
        MOV     R0,#54H    ;緩衝區已滿，緩衝區重設至開頭
        MOV     A,#C6H
        ACALL  W_INS
F48: SJMP    F41
;-----
; F4-SET;設定密碼
;-----
SETF4:  ACALL  W_CODE      ;將 4 位元組密碼資料寫入 93C66 中
        ACALL  CLRLCD      ;清除 LCD 螢幕
        MOV     A,#84H
        ACALL  W_INS
        MOV     CNT,#8
        MOV     DPTR,#MSG9 ;顯示'SUCCESS!'
        ACALL  W_STR

```

```

M0:    ACALL  SCAN      ;掃描鍵盤
        MOV     A,R5
        CJNE   A,#BH,M1 ;按下(HELP)鍵?
        AJMP   HELP
M1:    CJNE   A,#CH,M2 ;按下(F1)鍵?
        AJMP   F1
M2:    CJNE   A,#DH,M3 ;按下(F2)鍵?
        AJMP   F2
M3:    CJNE   A,#EH,M4 ;按下(F3)鍵?
        AJMP   F3
M4:    CJNE   A,#FH,M0 ;按下(F4)鍵?
        AJMP   F4
;-----
; SCAN KEY;掃描鍵盤
;-----
SCAN:   MOV     R5,#0      ;鍵值=0
        MOV     R7,#4      ;4 行
        MOV     R6,#11111110B ;行掃描信號
COL:    MOV     A,R6
        MOV     P3,A
        JB      P3.4,S1    ;第 1 列是否有按鍵按下?
        JNB     P3.4,$      ;除彈跳
        SJMP    S5
S1:     INC     R5          ;鍵值加 1
        JB      P3.5,S2    ;第 2 列是否有按鍵按下?
        JNB     P3.5,$      ;除彈跳
        SJMP    S5
S2:     INC     R5          ;鍵值加 1
        JB      P3.6,S3    ;第 3 列是否有按鍵按下?
        JNB     P3.6,$      ;除彈跳
        SJMP    S5
S3:     INC     R5          ;鍵值加 1
        JB      P3.7,S4    ;第 4 列是否有按鍵按下?
        JNB     P3.7,$      ;除彈跳
        SJMP    S5
S4:     INC     R5          ;鍵值加 1
        MOV     A,R6
        RL      A          ;行掃描信號左移 1 位元
        MOV     R6,A
        DJNZ    R7,COL     ;掃描下一行
S5:     RET
;-----
; POWER ON;顯示功能表
;-----
ON:     PUSH    A
        MOV     A,#80H      ;LCD 第 1 列第 1 行

```

```

ACALL W_INS
MOV     CNT,#16      ;顯示 16 個字
MOV     DPTR,#MSG1   ;顯示'F1:TIME F2:ADJ'
ACALL W_STR
MOV     A,#C0H       ;LCD 第 2 列第 1 行
ACALL W_INS
MOV     CNT,#16      ;顯示 16 個字
MOV     DPTR,#MSG2   ;顯示'F3:INPUT F4:SET'
ACALL W_STR
POP     A
RET

;-----
; SET TIMER0;計時器 0 設定
;-----
SET0:   MOV     SEC,#100      ;中斷次數
        MOV     TMOD,#01H    ;工作於模式 1
        MOV     IE,#10000010B ;致能計時器 0 中斷
        MOV     TH0,#>(65536-10000) ;每 10mS 中斷 1 次
        MOV     TL0,#<(65536-10000)
        SETB    TR0          ;啟動計時器 0
        RET

;-----
; INIT RAM & LCD;記憶體初值設定與 LCD 初始化
;-----
INIT:   MOV     R0,#6         ;6 位元組資料
        MOV     R1,#30H
        CLR     A
NEXT0:  MOV     @R1,A         ;清除 RAM 位址 30H~35H 內容
        INC     R1
        DJNZ    R0,NEXT0
        MOV     R0,#14       ;14 位元組資料
        MOV     R1,#40H
        CLR     A
NEXT1:  MOV     @R1,A         ;清除 RAM 位址 40H~4DH 內容
        INC     R1
        DJNZ    R0,NEXT1
;
        MOV     A,#38H       ;功能設定
        ACALL   W_INS
        MOV     A,#38H       ;功能設定
        ACALL   W_INS
        MOV     A,#38H       ;功能設定
        ACALL   W_INS
        MOV     A,#38H       ;功能設定
        ACALL   W_INS
        MOV     A,#08H       ;顯示器 OFF

```

```

ACALL W_INS
MOV     A,#01H      ;清除顯示
ACALL W_INS
MOV     A,#06H      ;輸入模式設定:游標右移，顯示字元不動
ACALL W_INS
MOV     A,#0CH      ;顯示器 ON
ACALL W_INS
RET

;-----
; CLEAR LCD;清除 LCD
;-----
CLRLCD: MOV     A,#01H
        ACALL W_INS
        RET

;-----
; BCD TO ASCII;時間值轉換:BCD 轉成 ASCII
;-----
CONV:  MOV     A,#30H      ;RAM 位址 30H:(時)BCD 資料
        MOV     R0,A
        MOV     A,@R0
        SWAP    A
        ANL     A,#0FH
        ADD     A,#30H
        MOV     40H,A      ;RAM 位址 40H:(十位時)ASCII 資料
        MOV     A,@R0
        ANL     A,#0FH
        ADD     A,#30H
        MOV     41H,A      ;RAM 位址 41H:(個位時)ASCII 資料
        MOV     42H,#":"
        INC     R0          ;RAM 位址 31H:(分)BCD 資料
        MOV     A,@R0
        SWAP    A
        ANL     A,#0FH
        ADD     A,#30H
        MOV     43H,A      ;RAM 位址 43H:(十位分)ASCII 資料
        MOV     A,@R0
        ANL     A,#0FH
        ADD     A,#30H
        MOV     44H,A      ;RAM 位址 44H:(個位分)ASCII 資料
        MOV     45H,#":"
        INC     R0          ;RAM 位址 32H:(秒)BCD 資料
        MOV     A,@R0
        SWAP    A
        ANL     A,#0FH
        ADD     A,#30H
        MOV     46H,A      ;RAM 位址 46H:(十位秒)ASCII 資料

```

```

        MOV      A,@R0
        ANL      A,#0FH
        ADD      A,#30H
        MOV      47H,A      ;RAM 位址 47H:(個位秒)ASCII 資料
        RET

;-----
; DISPLAY TIME;顯示時間於 LCD 上
;-----
DISP:   MOV      R1,#8      ;8 位元組時間資料
        MOV      R0,#40H   ;時間 ASCII 資料存於 RAM 位址 40H~47H 中
DSP:MOV  A,@R0
        ACALL    W_DATA
        INC      R0
        DJNZ     R1,DSP
        RET

;-----
; WRITE STRING TO LCD;寫入字串至 LCD
;-----
W_STR:  PUSH     A
        MOV      R5,#0
STR:MOV  A,R5
        MOVC     A,@A+DPTR
        ACALL    W_DATA
        INC      R5
        DJNZ     CNT,STR
        POP      A
        RET

;-----
; WRITE INS TO LCD;寫入指令至 LCD
;-----
W_INS:  PUSH     A
        ACALL    CHECK     ;檢查忙碌旗標
        CLR      RS        ;RS=0, 寫入指令至 LCD 模組
        CLR      RW
        SETB     EN        ;EN=1, 致能
        MOV      P1,A
        CLR      EN        ;EN=0, 除能
        POP      A
        RET

;-----
; WRITE DATA TO LCD;寫入資料至 LCD
;-----
W_DATA: PUSH     A
        ACALL    CHECK     ;檢查忙碌旗標
        SETB     RS        ;RS=0, 寫入資料至 LCD 模組
        CLR      RW

```

```

        SETB    EN            ;EN=1 , 致能
        MOV     P1,A
        CLR     EN            ;EN=0 , 除能
        POP     A
        RET

;-----
; CHECK LCD BUSY;檢查 LCD 忙碌旗標
;-----
CHECK:  PUSH    A
BUSY:   CLR     RS
        SETB    RW
        SETB    EN
        MOV     A,P1
        CLR     EN
        JB      ACC.7,BUSY
        ACALL   DELAY
        POP     A
        RET

;-----
; DELAY;延遲 6mS
;-----
DELAY:  MOV     R6,#20
D1:     MOV     R7,#150
        DJNZ    R7,$
        DJNZ    R6,D1
        RET

;-----
; JUDGE H/M/S;調整 時/分/秒
;-----
JUDGE:  PUSH    A
        MOV     A,32H
        ADD     A,#1
        DA      A
        MOV     32H,A
        CJNE    A,#60H,FIX    ;秒>60?
        MOV     32H,#00
        MOV     A,31H
        ADD     A,#1
        DA      A
        MOV     31H,A
        CJNE    A,#60H,FIX    ;分>60?
        MOV     31H,#00
        MOV     A,30H
        ADD     A,#1
        DA      A
        MOV     30H,A

```

```

        CJNE    A,#24H,FIX    ;時>24?
        MOV     30H,#00
FIX: POP     A
        RET
;-----
; TIMER0;計時器 0 中斷
;-----
TIMER0: PUSH    A
        MOV     TH0,#>(65536-10000)    ;每 10mS 中斷 1 次
        MOV     TL0,#<(65536-10000)
        DJNZ    SEC,EXIT3
        MOV     SEC,#100                ;中斷 100 次即 1 秒
        ACALL   JUDGE
EXIT3:  POP     A
        RETI
;-----
; READ CODE FORM 93CX6;自 93C66 讀取密碼資料
;-----
R_CODE: PUSH    A
        MOV     ADDR,#0
        MOV     R0,#54H                ;存入 RAM 位址 54H~57H 中
        MOV     R1,#2                  ;2 個 16 位元組資料
LOOP1:  ACALL   READ                    ;讀取
        INC     ADDR                    ;下一個位址
        DJNZ    R1,LOOP1
        POP     A
        RET
;-----
; WRITE CODE TO 93CX6;將密碼資料寫入 93C66 中
;-----
W_CODE: PUSH    A
        MOV     ADDR,#0
        MOV     R0,#54H
        MOV     R1,#2
LOOP2:  ACALL   ERASE                    ;清除
        ACALL   WEN                      ;寫入致能
        ACALL   WRITE                    ;寫入
        ACALL   WDS                      ;寫入除能
        INC     ADDR                    ;下一個位址
        DJNZ    R1,LOOP2
        POP     A
        RET
;-----
; ERASE 93CX6;清除
;-----
ERASE:  PUSH    A

```

```

MOV      R6,#00000111B
MOV      A,ADDR
MOV      R7,A
ACALL    SHIFT5
SETB     CS           ;CS=1
MOV      CNT,#11      ;11 位元指令
ER:      MOV      A,R7
        RLC      A
        MOV      R7,A
        MOV      A,R6
        RLC      A
        MOV      R6,A
        MOV      DI,C      ;串列資料移入 93C66 中
        SETB     SK        ;SK=1
        NOP
        CLR      SK        ;SK=0
        NOP
        DJNZ     CNT,ER
        ACALL    CHKBSY     ;檢查忙碌旗標
        CLR      CS        ;CS=0
        POP      A
        RET

```

```

;-----
; 93CX6 WRITE ENABLE;寫入致能
;-----

```

```

WEN:     PUSH     A
        MOV      R6,#00000100B
        MOV      R7,#11000000B
        ACALL    SHIFT5
        SETB     CS           ;CS=1
W1:      MOV      CNT,#11      ;11 位元指令
        MOV      A,R7
        RLC      A
        MOV      R7,A
        MOV      A,R6
        RLC      A
        MOV      R6,A
        MOV      DI,C      ;串列資料移入 93C66 中
        SETB     SK        ;SK=1
        NOP
        CLR      SK        ;SK=0
        NOP
        DJNZ     CNT,W1
        CLR      CS
        POP      A
        RET

```

```

;-----
; 93CX6 WRITE DISABLE;寫入除能
;-----
WDS:  PUSH  A
      MOV    R6,#00000100B
      MOV    R7,#00000000B
      ACALL  SHIFT5
      SETB   CS           ;CS=1
      MOV    CNT,#11      ;11 位元指令
DS:   MOV    A,R7
      RLC    A
      MOV    R7,A
      MOV    A,R6
      RLC    A
      MOV    R6,A
      MOV    DI,C         ;串列資料移入 93C66 中
      SETB   SK          ;SK=1
      NOP
      CLR    SK           ;SK=0
      NOP
      DJNZ   CNT,DS
      CLR    CS           ;CS=0
      POP    A
      RET

;-----
; WRITE;寫入
;-----
WRITE: PUSH  A
      MOV    R6,#00000101B
      MOV    A,ADDR
      MOV    R7,A
      ACALL  SHIFT5
      SETB   CS           ;CS=1
      MOV    CNT,#11      ;11 位元指令
WR:   MOV    A,R7
      RLC    A
      MOV    R7,A
      MOV    A,R6
      RLC    A
      MOV    R6,A
      MOV    DI,C         ;串列資料移入 93C66 中
      SETB   SK          ;SK=1
      NOP
      CLR    SK           ;SK=0
      NOP
      DJNZ   CNT,WR

```

```

        ACALL W_DAT
        ACALL CHKBSY      ;檢查忙碌旗標
        CLR    CS
        POP    A
        RET

;-----
; WRITE DATA TO 93CX6;將資料寫入 93C66 中
;-----
W_DAT:  PUSH    A
        MOV     R3,#2      ;2 位元組資料
WD1:    MOV     A,@R0
        MOV     R4,#8
WD2:    RLC     A
        MOV     DI,C
        SETB    SK         ;SK=1
        NOP
        CLR     SK         ;SK=0
        NOP
        DJNZ    R4,WD2
        INC     R0
        DJNZ    R3,WD1
        POP     A
        RET

;-----
; READ;讀取
;-----
READ:   PUSH    A
        MOV     R6,#00000110B
        MOV     A,ADDR
        MOV     R7,A
        ACALL   SHIFT5
        SETB    CS         ;CS=1
        MOV     CNT,#11    ;11 位元指令
RD:     MOV     A,R7
        RLC     A
        MOV     R7,A
        MOV     A,R6
        RLC     A
        MOV     R6,A
        MOV     DI,C      ;將串列資料移入 93C66 中
        SETB    SK         ;SK=1
        NOP
        CLR     SK         ;SK=0
        NOP
        DJNZ    CNT,RD
        ACALL   R_DAT

```

```

        CLR    CS            ;CS=0
        POP    A
        RET
;-----
; READ DATA FROM 93CX6;自 93C66 讀取資料
;-----
R_DAT:  PUSH    A
        MOV     R3,#2
RD1:    MOV     R4,#8
RD2:    SETB    SK            ;SK=1
        NOP
        CLR     SK            ;SK=0
        NOP
        MOV     C,DO          ;93C66 移出串列資料
        RLC     A
        DJNZ    R4,RD2
        MOV     @R0,A
        INC     R0
        DJNZ    R3,RD1
        POP     A
        RET
;-----
; SHIFT LEFT 5 BIT;左移 5 位元
;-----
SHIFT5: PUSH    A
        MOV     CNT,#5
SHT:    MOV     A,R7
        RLC     A
        MOV     R7,A
        MOV     A,R6
        RLC     A
        MOV     R6,A
        DJNZ    CNT,SHT
        POP     A
        RET
;-----
; CHECK 93CX6 BUSY;檢查 93C66 忙碌旗標
;-----
CHKBSY: PUSH    A
BSY:    CLR     CS            ;CS=0
        SETB    CS            ;CS=1
        MOV     C,DO          ;DO=0 , 93C66 忙碌中
        JNC     BSY           ;DO=1 , 93C66 閒置
        POP     A
        RET
;-----

```

```
; MESSAGE
;-----
MSG1:  DB      'F1:TIME  F2:ADJ'
MSG2:  DB      'F3:INPUT  F4:SET'
MSG3:  DB      'TIME'
MSG4:  DB      'ADJUST TIME'
MSG5:  DB      'INPUT CODE'
MSG6:  DB      'EXACT'
MSG7:  DB      'ERROR'
MSG8:  DB      'SET CODE'
MSG9:  DB      'SUCCESS!'
      END
```

陸、研究結果

經過這次的製作，我們了解到要維護電腦的安全並不是一件容易的事，但卻是一件相當重要且刻不容緩的事。

這次的製作中我們學到了很多也用到了一些以前沒有用過的電子零件，以及搭配單晶片的程式編碼、燒錄、模擬等。

也學到了「說的到不一定就做得得到」我們原本預計的期望是更高的，雖然學藝尚未完全成熟，但我們相信只要全組的組員都能夠盡全力的製作、相輔相成的心，以自己的長處來彌補別人不足的部分，使我們發輝團結的精神，相信一定能夠創造出最佳的成績。

而我們未來的期望是要將「電腦手護天使」能加以商品化後在加強其功能性，使以便更多的電腦使用者，可以放心的安裝自己專屬的個人電腦。

柒、討論

其實我們在製作的過程中也遇到了不少的難題。

- 1、密碼鎖的電路在麵包板上模擬較容易，但是在電路板上焊接時要接合鍵盤及液晶顯示器的接線過於繁瑣，接線不易。
- 2、在密碼鎖電路與電腦主機結合時的問題較麻煩，可能因為出錯，而損壞到電腦主機。
- 3、這次所使用到的電子零件其中有部分的腳位過多或著是過大等問題，容易造成焊接不易的問題。
- 4、有一些電子零件的腳位不只一種規格，容易造成設計佈線圖的不易，例如：LCD 液晶顯示器。

捌、結論

上面那些問題的解決方式如下：

- 1、解決的方法就是先將鍵盤焊接在另一塊電路板上在以牽線的方式至密碼鎖的電路板上即可，而液晶顯示器則以排線連接即可。
- 2、解決的方法就是從較簡易的開關接線部分截入，用以直接控制電腦的電源，也較不會妨礙到電腦的正常運作。
- 3、解決的方法就是，有母座的電子零件就盡量使用母座，沒有的就只好自己擴孔了。
- 4、關於液晶顯示器腳位的詳細資料，只要查閱相關的工具書即可查道詳細的資料，方便佈線。

玖、參考資料及其它

作者：楊明豐

書名：8051 單晶片設計實務

版次：初版

出版地：台北市南港路 3 段 50 巷 7 號 5 樓

出版社：碁峰資訊股份有限公司

頁數：第十七章全

出版年：89 年 9 月

作者：蔡朝洋

書名：8051/52 單晶片微電腦原理與應用

版次：三版十刷

出版地：台北市龍江路 76 巷 20-2 號 2 樓

出版社：全華科技圖書股份有限公司

頁數：p423~p427

出版年：87 年 12 月